

```
"NAME"
"IDEAL n of 1 sample size V1"

"PURPOSE"
"Sample size calculation for n of 1 trials"
"*****"

"FUNDING"
"This work was partly supported by the European Union's 7th Framework Programme
for research, technological development and demonstration under grant agreement no. 602552, IDEAL"
"*****"

"AUTHOR"
"Stephen Senn"
"*****"

"EXPLANATION"

"The purpose of the program is to provide sample size calculations for trials with two treatments
(say A and B) randomised in k cycles. In each cycle it is assumed that each patient will
receive one treatment with A and one with B in random order.

Outcome measures are asumed to be (approximately) Normally distributed

Three fundamental tasks are addressed.
I) Testing the overall effect of treatment across patients
II) Providing estimates of the treatment effect in a given patient
III) Considering the variance of the ratio of the weights that would be used for shrunk estimates

For the first task, two basic types of power calculation are provided.
1) Fixed effects calculation (calculation I.1)
2) Random effects calculation (calculation I.2)

For both these cases the sample size is rounded up to the nearest integer
providing the requested power. The associated power for this sample size
is also provided.

For the second task, two approaches are covered.
1) Naive calculation using just the results for that patient (calculation II.1)
2) Shrunk estimation using a combination of inidividual results for (calculation II.2)
that patient and overall results

For both these csaes the standard errors that would apply for the parameter settinsg chosen are
provided.

For the third task, the standard error of the ratio of the weight for the overall mean to the
weight for the
patients mean is calculated as a function of the overall sample size (calculation III.1)

For the second task and the second approach (II.2) it is assumed that the various components of
variance
and the overall mean have been estimated with high precision and that uncertainty as regards their
values is unimportant

For these five cases, solutions are provided for various values of k

The user must put in the maximum number of cycles for which power and sample size will be
determined
This and other parameters must be set in the section headed INPUT

The theory for all these cases is covered in the paper:
Sample size considerations for n-of-1 trials, by Stephen Senn "
"*****"

"BEGIN"
"Start program"
OUTPUT[PRINT=*"Handles pagination"

"*****"
"INPUT"
```

"The parameter values in this section must be set by the user substituting his or her own values where it says VALUE= "

"Choose whether variances and standard errors of ratios will be printed or not.
(Printing them takes up a lot of space)"

SCALAR [VALUE= 0]noyes1 "0= No, 1=Yes"

"Choose whether plots will be in colour or not"

SCALAR[VALUE=1]noyes2 "0= No, 1=Yes"

"Set parameter values for calculation"

"NB CycleMax sets the maximum value of k that will be considered for sample size calculation
(Tasks I.1, I.2, II.1 & II.3)

KExamp sets the number of cycles considered for calculating the standard error of the ratio of weights
(Task III)"

"Maximum number of cycles. Must be Integer"

SCALAR[VALUE=15;IPRINT=EXTRA]CycleMax;EXTRA='Maximum number of cycles per patient';DECIMALS=0

SCALAR[VALUE=3;IPRINT=EXTRA]KExamp;DECIMALS=0;\

EXTRA='Number of cycles for plotting standard error of ratio'; "NB must be integer > 1"

"Parameters governing relative precision"

SCALAR[VALUE=1;IPRINT=EXTRA]Delta;EXTRA='~{Delta} Clinically relevant difference'

SCALAR[VALUE=1;IPRINT=EXTRA]Psi;EXTRA=' ~{Psi} SD of treatment-by-patient interaction'

SCALAR[VALUE=2;IPRINT=EXTRA]Sigma;EXTRA=' ~{sigma} Within cycle SD' "SD of error from occasion to occasion"

"Parameters governing type I and II errors"

SCALAR[VALUE=0.05;IPRINT=EXTRA]Alpha;EXTRA=' ~{alpha} Type I error rate'

SCALAR[VALUE=0.2;IPRINT=EXTRA]Beta;EXTRA=' ~{beta} Type II error rate'

SCALAR[VALUE=2;IPRINT=EXTRA]Sides;EXTRA=' sides for test';DECIMALS=0;"Integer = 1 or 2"

"Parameters governing extent of output"

SCALAR[VALUE=20;IPRINT=EXTRA]MinN;DECIMALS=0;\

EXTRA='Minimum number of patients for calculating variance of ratio';"Integer NB must be >=5"

SCALAR[VALUE=100;IPRINT=EXTRA]MaxN;DECIMALS=0;\

EXTRA='Maximum number of patients for calculating variance of ratio'; "NB must be >= MinN"

"End INPUT"

"Terminate program if inputs incorrect"

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='Sides for test must be either 1 or 2'] (Sides.NE.1).AND.
(Sides.NE.2)

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='Alpha must be greater than 0 and less than 1']
(Alpha.LE.0).OR.(Alpha.GE.1)

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='Beta must be greater than 0 and less than 1']
(Beta.LE.0).OR.(Beta.GE.1)

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='Sigma & Psi must both be strictly positive']
(Sigma.LE.0).OR.(Psi.LE.0)

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='MaxN cannot be less than MinN'] (MaxN.LT.MinN)

EXIT[CONTROLSTRUCTURE=Job;EXPLANATION='CycleMax must be integer greater than 1']\
(CycleMax.LT.2).OR.(Floor(CycleMax).LT.CycleMax)

"CALCULATION"

"Begin calculation"

"Calculate target power"

SCALAR Power,TPower; "Target power (prob) & Target power %"

SCALAR Test_Alpha "Calculate one sided value corresponding to sides chosen"

SCALAR[IPRINT=EXTRA]ABDelta;EXTRA='Abs(~{Delta}) '

SCALAR Length1 "Length of vectors"

CALCULATE Length1=CycleMax-1

CALCULATE Power=1-Beta

```
CALCULATE TPower=100*Power
CALCULATE ABDelta=ABS(Delta)
CALCULATE Test_Alpha=Alpha/Sides
TXCONSTRUCT [TEXT=Label]'Target power is ',TPower,'%'
```

"Declare various variates"

```
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FVar;EXTRA='Variance' "Fixed effects variance"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]RVar;EXTRA='Variance' "Random effects variance"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FN;DECIMALS=0;EXTRA='Patients' "Number of patients fixed
effects I.1"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]RN;DECIMALS=0;EXTRA='Patients' "Number of patients random
effects I.2"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]TCrit;EXTRA='T critical value' "Critical value fixed effects
I.1"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FPowerE;EXTRA='Exact Power' "Fixed effects power (exact)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]RPower;EXTRA='Power' "Random effects power"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FTotal;DECIMALS=0;EXTRA='Total no of cycles' "Fixed effects
case I.1 (approximate)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FTotal2;DECIMALS=0;EXTRA='Total no of cycles' "Fixed effects
case I.1 (exact)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]RTotal;DECIMALS=0;EXTRA='Total no of cycles' "Random effects
case II.1"
VARIATE[VALUES=2...#CycleMax;IPRINT=EXTRA]nCycles;DECIMALS=0;EXTRA='Cycles, ~i{k}' "Number of
cycles one has decided to use"
VARIATE[VALUES=1...#CycleMax;IPRINT=EXTRA]nCycles2;DECIMALS=0;EXTRA='Cycles, ~i{k}' "Number of
cycles one has decided to use"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FDF;DECIMALS=0;EXTRA='Fixed effects DF' "DF fixed effects
I.1"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]Noncent;EXTRA='Non-centrality' "Non-centrality for t"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FDF2;DECIMALS=0;EXTRA='Fixed effects DF' "DF fixed effects
I.1 (for using during search)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]Noncent2;EXTRA='Non-centrality' "Non-centrality for t (for
use during search)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FN2;DECIMALS=0;EXTRA='Patients (Exact)' "Number of patients
fixed effects I.1 (exact)"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]FPowerE2;EXTRA='Exact Power' "Fixed effects power (exact) for
iteration"
VARIATE[NVALUES=Length1;IPRINT=EXTRA]TCrit2;EXTRA='T critical value' "Critical value fixed effects
I.1 (for use during search)"
```

"Declare various temporary scalars"

```
SCALAR K,Loop,FNTemp,FN2Temp,DFTemp;DECIMALS=0
SCALAR Zalpha,Zbeta,FPTemp,NoncentT,CritTemp
CALCULATE Zalpha=ABS(EDNORMAL(Test_Alpha))
CALCULATE Zbeta=ABS(EDNORMAL(Beta))
```

FOR[INDEX=I;NTIMES=Length1] "Begin cycle loop"

```
    CALCULATE K=I+1
    CALCULATE FVar$[I]=2*(Sigma**2)/K "Variance of patient contrast fixed effects"
    "Begin fixed effects sample size search"
    "Begin calculation of temporary values based on Normal distribution"
    CALCULATE FNTemp=CEILING((2*(Sigma**2)*(Zalpha+Zbeta)**2)/(K*(ABDelta)**2)) "Sample size for
Normal"
    CALCULATE DFTemp=FNTemp*(K-1)
    "Use these values to calculate statistics for t-distribution"
    CALCULATE CritTemp=ABS(EDT(Test_Alpha;DFTemp;0))
    CALCULATE NoncentT=ABDelta/SQRT((2*Sigma**2)/(FNTemp*K))
    CALCULATE FPTemp=CUT(CritTemp;DFTemp;NoncentT)
    "PRINT FNTemp,DFTemp,CritTemp,NoncentT,FPTemp"
    "Start assigning these temporary values as initial ones for vector elements"
    CALCULATE FN$[I]=FNTemp "Store approximate sample size and power"
    CALCULATE FDF$[I]=DFTemp "Store degrees of freedom for within-patient variance"
    CALCULATE Noncent$[I]=NoncentT "Calculate non-centrality"
    CALCULATE TCrit$[I]=CritTemp "Calculate critical value"
    CALCULATE FPowerE$[I]=100*FPTemp "Calculate exact power"
    "Begin iterative search for exact sample size"
    "Calculate search over next 5 sample sizes"
```

```
    FOR [INDEX=J;NTIMES=5] "Begin search loop"
```

```

    CALCULATE FN2Temp=FN$[I]+J-1 "Start with value indicated by Normal"
    CALCULATE FDF2$[I]=FN2Temp*(K-1)"Calculate degrees of freedom"
    CALCULATE Noncent2$[I]=ABDelta/SQRT((2*Sigma**2)/(FN2Temp*K)) "Calculate non-centrality"
    CALCULATE TCrit2$[I]=ABS(EDT(Alpha/Sides;FDF2$[I];0)) "Calculate critical value"
    CALCULATE FPowerE2$[I]=100*CUT(TCrit2$[I];FDF2$[I];Noncent2$[I]) "Calculate exact power"
    EXIT[CONTROLSTRUCTURE=FOR]FPowerE2$[I].GT.TPower "Exit once power greater than target
power"
    ENDFOR "End search loop"

    CALCULATE FN2$[I]=FN2Temp "Store sample size"

    "Begin random effects sample size calculation"
    CALCULATE RVar$[I]=(Psi**2)+2*(Sigma**2)/K "Variance of patient contrast random effects"
    STTEST [NSAMPLES=1;TMETHOD=onesided;POWER=Power;PROBABILITY=Test_Alpha;PRINT=*]
RESPONSE=ABDelta; \
    VAR1=RVar$[I]; NREPLICATES=RNTemp;VPOWER=RPTemp "Sample size RNTEMP and associated power
RPTEMP"
    CALCULATE RN$[I]=RNTemp "Store sample size and power"
    CALCULATE RPower$[I]=100*RPTemp$[6] "Calculate and store power"
    ENDFOR "End cycle loop"

CALCULATE FTotal,RTotal,FTotal2=nCycles*FN,RN,FN2 "Calculate total number of cycles for I.1 & I.2"

"Now calculate naive and shrunk standard errors as functions of the number of cycles"
VARIATE [NVALUES=CycleMax;IPRINT=EXTRA]Naive,Shrunk;EXTRA='Naive SE','Shrunk SE'
CALCULATE Naive=SQRT(2*(Sigma**2)/nCycles2) "Task II.1"
CALCULATE Shrunk=SQRT((2*(Sigma**2)*(Psi**2))/(nCycles2*(Psi**2)+2*(Sigma**2))) "Task II.2"

"To calculate the variance of the ratios of weights used in producing shrunk estimates"
VARIATE [VALUES=2...CycleMax] Suffs

POINTER[SUFFIXES=Suffs;IPRINT=EXTRA]VR,SER,ratio;\
EXTRA='Variance of the ratio of weights','SE of the ratio of weights','ratio of weights'

VARIATE [VALUES=#MinN...#MaxN;IPRINT=EXTRA]nVR;\
EXTRA='number of patients';DECIMALS=0
SCALAR Length2;EXTRA='Number of sample sizes'
CALCULATE Length2=NVALUES(nVR)
VARIATE [NVALUES=#Length2]VR[],SER[],ratio[]
FOR [INDEX=K;START=2;END=CycleMax]
    CALCULATE VR[K]=(((K*Psi**2)/(2*Sigma**2))+1)**2*\
        (2*(nVR*(K-1))**2)*(nVR*K-3)/\
        ((nVR-1)*((nVR*(K-1)-2)**2)*(nVR*(K-1)-4))
    CALCULATE ratio[K]=Psi**2/(2*(Sigma**2)/K)
ENDFOR
CALCULATE SER[]=SQRT(VR[])

"*****"
"OUTPUT"
"Print design parameters chosen"
PAGE
CAPTION 'Power calculations for n of 1 trials';STYLE=Major
CAPTION 'Author: Stephen Senn';STYLE=Stress
CAPTION 'This work was partly supported by the European Union's 7th Framework Programme \
for research, technological development and demonstration under grant agreement no. 602552,
IDEAL';\
STYLE=Plain

PRINT ''
PRINT ''

CAPTION 'Program settings';STYLE=Minor
CAPTION 'Parameters';STYLE=STRESS
PRINT CycleMax, KExamp
PRINT Alpha,Beta
PRINT Delta
PRINT Psi, Sigma

```

```
PRINT Sides
PRINT MinN,MaxN
```

PAGE

"Print results"

CAPTION 'Power calculations for testing overall effect of treatment';STYLE=Minor

CAPTION Label

CAPTION 'Variance quoted is for the treatment effect per patient';STYLE=Note

CAPTION 'Power calculations for fixed effects case';STYLE=Stress

CAPTION 'Variance = $2\sigma^2/\sim{i}{k}$ ';STYLE=Plain

CAPTION 'Approximate solution based on simple Normal formula for sample size ';STYLE=NOTE

PRINT nCycles,FVar,FN,FTotal,FPowerE

CAPTION 'Exact solution';STYLE=NOTE

PRINT nCycles,FVar,FN2,FTotal2,FPowerE2

CAPTION 'Power calculations for random effects case';STYLE=Stress

CAPTION 'Variance = $\sim{\psi}^2+2\sigma^2/\sim{i}{k}$ ';STYLE=Plain

PRINT nCycles,RVar,RN,RTotal,RPower

PAGE

CAPTION 'Expected precision of predictions';STYLE=Minor

CAPTION 'Standard errors of predictions for individual patients';STYLE=Stress

CAPTION 'Naive $\sim{\sqrt{2\sigma^2/\sim{i}{k}}}$ '

CAPTION 'Shrunk $\sim{\sqrt{2\sigma^2\sim{\psi}^2/(\sim{i}{k}\sim{\psi}^2+2\sigma^2)}}$ '

CAPTION 'Random effects model assumed';STYLE=Note

PRINT nCycles2,Naive,Shrunk

IF noyes1==1

PAGE

CAPTION 'Variance of ratio of weights';STYLE=Minor

PRINT nVR, VR[]

PRINT nVR, SER[]

ENDIF

CAPTION 'End of output';STYLE=Stress

"*****"

"PLOTING"

"Some calculations for plots"

CALCULATE ratio_ref=Psi**2/(2*(Sigma**2)/KExamp)"Reference ratio of weights for plotting"

SCALAR Yupper

IF ratio_ref.GT.SER[KExamp]\$[1]

CALCULATE Yupper=1.05*ratio_ref

ELSE

CALCULATE Yupper=1.05*SER[KExamp]\$[1]

ENDIF

"Define positions for plotting text"

SCALAR X,Y

CALCULATE X=4;Y=RN\$[2]

"Define pens for plotting"

IF noyes2==0

PEN [RESET=yes] 1...4; METHOD=Line; SYMBOL=2,3,0,0; JOIN=ascending; LINESTYLE=1,1,1,3;\

COLOUR='Black','Grey','Black','Grey'

ELSE

PEN [RESET=yes] 1...4; METHOD=Line; SYMBOL=2,3,0,0; JOIN=ascending; LINESTYLE=1,1,1,3;\

COLOUR='Blue','Red','Black','Black'

ENDIF

"Plot sample size in terms of patients"

FRAME [RESET=yes] WINDOW=1; XLOWER=0; XUPPER=0.75; YLOWER=0.25; YUPPER=1; XMLOWER=0.12;\

XMUPPER=0.05; YMLOWER=0.1; YMUPPER=0.07; BOX=omit

XAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=parallel; MPOSITION=outside;\

ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;TITLE='Cycles per patient'

YAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=perpendicular; MPOSITION=outside;\

```

ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;LOWER=0;TITLE='Number of patients'
DGRAPH [WINDOW=1;TITLE='Sample size calculation for n of 1 trials'] Y=FN2,RN; X=2(nCycles);\
DESCRIPTION='Fixed Effect', 'Random Effect';PEN=1,2
DTEXT TEXT=Label;X=X;Y=Y;PEN=3

```

"Plot sample size in terms of total numbers of cycles"

```

FRAME [RESET=yes] WINDOW=1; XLOWER=0; XUPPER=0.75; YLOWER=0.25; YUPPER=1; XMLOWER=0.12;\
XMUPPER=0.05; YMLOWER=0.1; YMUPPER=0.07; BOX=omit
XAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=parallel; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;TITLE='Cycles per patient'
YAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=perpendicular; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;LOWER=0;TITLE='Total number of cycles'
DGRAPH [WINDOW=1;TITLE='Sample size calculation for n of 1 trials'] Y=FTotal2,RTotal; X=2
(nCycles);\
DESCRIPTION='Fixed Effect', 'Random Effect';PEN=1,2
DTEXT TEXT=Label;X=X;Y=Y;PEN=3

```

"Plot SEs of naive and shrunk estimates"

```

FRAME [RESET=yes] WINDOW=1; XLOWER=0; XUPPER=0.75; YLOWER=0.25; YUPPER=1; XMLOWER=0.12;\
XMUPPER=0.05; YMLOWER=0.1; YMUPPER=0.07; BOX=omit
XAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=parallel; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;TITLE='Number of cycles'
YAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=perpendicular; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;LOWER=0;TITLE='Standard error'
DGRAPH [WINDOW=1;TITLE='Standard errors for individual estimates']\
Y=Naive,Shrunk; X=2(nCycles2);DESCRIPTION='Naive', 'Shrunk' ;PEN=1,2
DREFERENCELINE Psi;LABEL='SE using previous data only';XLPOSITION=Right;YLPOSITION=above;PEN=4

```

"Plot SE of ratio of weights"

```

FRAME [RESET=yes] WINDOW=1; XLOWER=0; XUPPER=0.75; YLOWER=0.25; YUPPER=1; XMLOWER=0.12;\
XMUPPER=0.05; YMLOWER=0.1; YMUPPER=0.07; BOX=omit
XAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=parallel; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;TITLE='Number of patients'
YAXIS [RESET=yes] WINDOW=1; LPOSITION=outside; LDIRECTION=perpendicular; MPOSITION=outside;\
ARROWHEAD=omit; ACTION=display; TRANSFORM=identity;LOWER=0;TITLE='Standard error';\
UPPER=Yupper
DGRAPH [WINDOW=1;TITLE='Standard errors for ratios of weights']\
Y=SER[#KExamp]; X=nVR;DESCRIPTION='SE of ratio' ;PEN=1
DREFERENCELINE ratio_ref;LABEL='Ratio of weights';XLPOSITION=Right;YLPOSITION=above;PEN=4

```